

QUADRATIC TREND SURFACE

MAP = 26.2 X VALUE RIGHT EDGE OF MAP = 12.0 Y VALUE TOP EDGE OF MAP = .0  
REFERENCE CONTOUR VALUE = -2000.0 CONTOUR INTERVAL = 100.0



# Balgol Program for Trend-Surface Mapping Using an IBM 7090 Computer

By  
John W. Harbaugh  
Stanford University

SPECIAL DISTRIBUTION PUBLICATION 3

State Geological Survey  
The University of Kansas, Lawrence, Kansas  
1963

STATE GEOLOGICAL SURVEY OF KANSAS

W. Clarke Wescoe, M.D., Chancellor of The University and ex officio Director of the Survey

Frank C. Foley, Ph.D., State Geologist and Director

William W. Hambleton, Ph.D., Assoc. State Geologist and Assoc. Director

Raymond C. Moore, Ph.D., Sc.D., Principal Geologist Emeritus

John M. Jewett, Ph.D., Senior Geologist

Doris E. Nodine Zeller, Ph.D., Geologist, Editor

Grace M. Muilenburg, B.S., Public Information Director

Kenneth J. Badger, Chief Draftsman

Lila M. Watkins, Secretary

Research Divisions

Basic Geology.....Daniel F. Merriam, Ph.D., geologist in charge

Petrography and Geochemistry.....Ada Swineford, Ph.D., petrographer in charge

Mineral Resources.....Allison L. Hornbaker, M.S., geologist in charge

Oil and Gas.....Edwin D. Goebel, M.S., geologist in charge

Ceramics.....Norman Plummer, A.B., ceramist in charge

Cooperative Studies with United States Geological Survey

Ground-Water Resources.....Robert J. Dingman, B.S., geologist in charge

Mineral Fuels.....William D. Johnson, Jr., B.S., geologist in charge

Branch Offices

Geological Survey Well Sample Library, 4150 Monroe, Wichita

Geological Survey Southeast Kansas District Field Office, College Station, Pittsburg

Geological Survey Southwest Kansas Field Office, 310 N. 9th Street, Garden City

## PREFACE

This special publication describes a computer program that will be useful to geologists in Kansas and elsewhere; it is the first of a series of publications of the Kansas Geological Survey in which the objective is to present details of computer programs that should be of general usefulness to geologists and petroleum engineers. The computer revolution is fast sweeping through the petroleum industry, but many small companies and independent operators do not have employees on their staffs who are skilled in computer applications. It is the intention of the Kansas Geological Survey to provide some assistance in computer applications in solving geological and petroleum engineering problems.

## INTRODUCTION

The program described here is for trend-surface mapping of geological data, such as geologic structure, and for smoothing geophysical data. A trend surface is a plane or smoothly curving surface that is fitted so that it passes above some data points and below others. The difference between the trend-surface value and the actual data point value is called the deviation or residual value. Trend surfaces are generally fitted so that the sum of the squared deviations is the least possible and may be termed least-squares trend surfaces. A more complete discussion of the applications of least-squares trend surfaces is given by Forgotson (1963) and Merriam and Harbaugh (in press), and the principles have been discussed by Krumbein (1959) and Harbaugh (in press).

The program for fitting first-, second-, and third-degree trend surfaces is written in BALGOL and may be used on any IBM 7090/1401 computer system for which a 7090 BALGOL compiler tape is available. The program was developed at the Computation Center at Stanford University. Peter Carah wrote that part of the program dealing with plotting of residual values, and procedure SOLV was written by the Stanford Computation Center staff.

The Kansas Geological Survey will distribute the program in punched card form for a limited time at a cost of \$10. The 7090 BALGOL compiler tape may be obtained by sending a magnetic tape to the Computation Center, Stanford University, Stanford, California.

The trend-surface program is fast and efficient, compiling in 18 seconds from the BALGOL source deck, requiring about 30 seconds to execute for a data set consisting of about 200 data points, and producing printed maps of about 60 square inches each. The data points may be irregularly spaced. The program prints contoured trend-surface maps and plots original and residual values on maps that are ready to be contoured manually. The program may be used with any number of sets of data. Many of the steps in the program are similar or identical to a four-variable trend-surface program described by Harbaugh (in press).

A manual describing BALGOL and entitled "Burroughs Algebraic Compiler, Revised Edition" (1963) may be obtained from the Burroughs Corporation, Detroit 32, Michigan. "An Introduction to BALGOL" (1961) by R. V. Oakford and J. M. Gere is available through the Stanford University Bookstore.

## STEPS IN TREND-SURFACE PROGRAM

Steps in the trend-surface program are outlined in the flow chart (Fig. 1) and are listed in detail in Table 1. Each line (card) of the program is numbered, permitting parts of the program to be referred to by the line numbers.

Steps in the program are tabulated below:

- (1) Integer declarations: lines 7-8.
- (2) Dimensions of arrays: lines 9-12.
- (3) Assign characters to alphanumeric arrays used for headings and for contoured maps: lines 13-21.
- (4) Format statements: lines 22, 24, 225-230, 232-241, 245, 246, 251-260, 262, 264, 266, 292-299, 308-313, 326.
- (5) Input statements: lines 26, 28, 31-41.
- (6) Clear arrays reserved for matrices to zero: lines 42-43.
- (7) Generate matrix and column vector for solution of normal equations (for discussion see Harbaugh, in press): lines 46-113.
- (8) Assign linear, linear plus quadratic, and linear plus quadratic plus cubic matrices and column vectors to new matrices and column vectors before inversion: lines 114-122.
- (9) Declare external procedure INPROD (used in matrix inversion procedure): line 124.
- (10) Declare procedure SOLV (for inversion of matrix): lines 125-173.
- (11) Call and execute procedure SOLV for inverting matrices of first-, second-, and third-degree surfaces: lines 174-178.
- (12) Calculate trend and residual values for all data points: lines 180-194.
- (13) Calculate error measure: lines 197-206.
- (14) Calculate sum of squares of trend values and residuals: lines 207-218.
- (15) Calculate percent of total sum of squares: lines 219-223.
- (16) Output statements: lines 224, 231, 243, 244, 247-250, 263, 267, 291, 307, 390, 399.
- (17) Calculate arithmetic mean: line 242.
- (18) Write statements: lines 261, 265, 268, 269, 300-302, 321, 323, 324, 334, 336, 337, 353, 356, 357, 392, 407, 448.
- (19) Calculate volume beneath first-, second-, and third-degree trend surfaces by double integration: lines 273-287.
- (20) Calculate spatially weighted averages by dividing trend-surface volumes by area beneath trend surface: lines 288-290.
- (21) Calculate horizontal and vertical increment values for printing of maps: lines 305-306.
- (22) Contour first-, second-, and third-degree trend-surface maps by successively evaluating X and Y coordinate values: lines 314-379.




Figure 1. Simplified flow chart of major steps in trend-surface program.

Table 1.--Listing of BALGOL statements in trend-surface program. Each line represents one punched card of program. Numbers in left column are for reference purposes and in practice must be placed in columns 73-80 on punched cards. Each program card must have the number 2 punched in column 1.

```

1 COMMENT PROGRAM TO FIT LINEAR, LINEAR + QUADRATIC, AND LINEAR + QUAD-
2 RATIC + CUBIC TREND SURFACES WHERE X AND Y VALUES ARE IRREGULARLY
3 SPACED. PROGRAM PRINTS OUT Z VALUES CALCULATED AT THE GIVEN X-Y POINTS
4 AND ALSO COMPUTES DEVIATIONS OF ACTUAL Z VALUES FROM TREND VALUES.
5 ERROR MEASURE AND PERCENT TOTAL SUM OF SQUARES ARE CALCULATED. TREND
6 SURFACES ARE CONTOURED. J. W. HARBAUGH, GEOLOGY DEPT., STANFORD $
7 INTEGER XP(I), I, J, K, L, OP, N, VERT, HOR, CV(I), M, W $
8 INTEGER DIGITS, PLTAR, PRINT, Q, IY, IX, LINE, TEMP, ARMIN, ARPLS $
9 ARRAY X(500,9), T(10,10), R(10), XP(500,3), T3(5,5), T6(8,8),
10 T10(12,12), R3(3), R6(6), R10(10), S3(3), S6(6), S10(10),
11 Z(132), CV(132) $
12 ARRAY PRINT(3,50), IY(500), IX(500), DIGITS(50) $
13 ARRAY PLTAR(12)=(1Z VALU, 1ES ORI, 1G DATA, 1ST OR,
14 1DER RE, 1SIDUAL, 12ND OR, 1DER RE, 1SIDUAL, 13RD OR,
15 1DER RE, 1SIDUAL) $
16 ARRAY ARMIN(40) = (1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
17 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
18 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
19 ARRAY ARPLS(40) = (1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
20 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
21 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1,
22 FORMAT FORM($K$(B$PRINT(3,I)$, $I$ DIGITS(I)$), W) $
23 OUTPUT PLOT(PLTAR(1.5Q - 3.5), PLTAR(1.5Q - 2.5), PLTAR(1.5(Q-1))) $
24 FORMAT PLFT(3A6, W, W) $
25 START..
26 INPUT ALPH(A1, A2, A3, A4, A5, A6, A7, A8, A9, A10, A11, A12) $
27 READ ($$ ALPH) $
28 INPUT OPT( OP, N, VERT, HOR, XR, XL, YB, YT, RF, CON) $
29 READ ($$ OPT) $
30 HAR = HOR $
31 IF OP EQL 2222 $
32 BEGIN
33 INPUT DATI(FOR I =(1,1,N) $ FOR J =(1,1,3) $ XP(I,J)) $
34 READ ($$ DATI) $
35 FOR I =(1,1,N) $ FOR J =(1,1,3) $ X(I,J) = XP(I,J) $
36 END $
37 IF OP EQL 4444 $
38 BEGIN
39 INPUT DATD(FOR I =(1,1,N) $ FOR J =(1,1,3) $ X(I,J)) $
40 READ ($$ DATD) $
41 END $
42 FOR K =(1,1,10) $ FOR L =(1,1,10) $ T(K,L) = 0.0 $
43 FOR K =(1,1,10) $ R(K) = 0.0 $
44 COMMENT CALCULATE VALUES FOR 10 X 10 T MATRIX ELEMENTS, EXCEPT FOR
45 THOSE VALUES THAT ARE DUPLICATES OF OTHERS, WHICH WILL BE ASSIGNED $
46 FOR I =(1,1,N) $
47 BEGIN
48 T(1,2) = T(1,2) + X(I,1) $
49 T(1,3) = T(1,3) + X(I,2) $
50 T(1,4) = T(1,4) + (X(I,1)*2.0) $
51 T(1,5) = T(1,5) + (X(I,1).X(I,2)) $
52 T(1,6) = T(1,6) + (X(I,2)*2.0) $
53 T(1,7) = T(1,7) + (X(I,1)*3.0) $
54 T(1,8) = T(1,8) + ((X(I,1)*2.0).(X(I,2))) $
55 T(1,9) = T(1,9) + ((X(I,1)).(X(I,2)*2.0)) $
56 T(1,10) = T(1,10) + (X(I,2)*3.0) $
57 T(2,7) = T(2,7) + (X(I,1)*4.0) $
58 T(2,8) = T(2,8) + ((X(I,1)*3.0).(X(I,2))) $
59 T(2,9) = T(2,9) + ((X(I,1)*2.0).(X(I,2)*2.0)) $
60 T(2,10) = T(2,10) + ((X(I,1)).(X(I,2)*3.0)) $
61 T(3,10) = T(3,10) + (X(I,2)*4.0) $
62 T(4,7) = T(4,7) + (X(I,1)*5.0) $
63 T(4,8) = T(4,8) + ((X(I,1)*4.0).(X(I,2))) $
64 T(4,9) = T(4,9) + ((X(I,1)*3.0).(X(I,2)*2.0)) $
65 T(4,10) = T(4,10) + ((X(I,1)*2.0).(X(I,2)*3.0)) $
66 T(5,10) = T(5,10) + ((X(I,1)).(X(I,2)*4.0)) $
67 T(6,10) = T(6,10) + (X(I,2)*5.0) $
68 T(7,4) = T(7,4) + (X(I,1)*5.0) $
69 T(7,7) = T(7,7) + (X(I,1)*6.0) $
70 T(7,8) = T(7,8) + ((X(I,1)*5.0).(X(I,2))) $

```

```

71 T(7,9) = T(7,9) + ((X(I,1)*4.0).(X(I,2)*2.0)) $
72 T(7,10) = T(7,10) + ((X(I,1)*3.0).(X(I,2)*3.0)) $
73 T(8,10) = T(8,10) + ((X(I,1)*2.0).(X(I,2)*4.0)) $
74 T(9,10) = T(9,10) + ((X(I,1)).(X(I,2)*5.0)) $
75 T(10,10) = T(10,10) + (X(I,2)*6.0) $
76 R(1) = R(1) + X(I,3) $
77 R(2) = R(2) + ((X(I,1)).(X(I,3))) $
78 R(3) = R(3) + ((X(I,2)).(X(I,3))) $
79 R(4) = R(4) + ((X(I,1)*2.0).(X(I,3))) $
80 R(5) = R(5) + ((X(I,1)).(X(I,2).X(I,3))) $
81 R(6) = R(6) + ((X(I,2)*2.0).(X(I,3))) $
82 R(7) = R(7) + ((X(I,1)*3.0).(X(I,3))) $
83 R(8) = R(8) + ((X(I,1)*2.0).(X(I,2).X(I,3))) $
84 R(9) = R(9) + (X(I,1).(X(I,2)*2.0).(X(I,3))) $
85 R(10) = R(10) + ((X(I,2)*3.0).(X(I,3))) $
86 END $
87 T(1,1) = N $
88 T(2,1) = T(1,2) $ T(5,2) = T(1,8) $
89 T(3,1) = T(1,3) $ T(5,4) = T(2,8) $ T(7,3) = T(2,8) $
90 T(4,1) = T(1,4) $ T(2,2) = T(1,4) $ T(5,5) = T(2,9) $ T(6,4) = T(2,9) $
91 T(5,1) = T(1,5) $ T(2,3) = T(1,5) $ T(3,2) = T(1,5) $ T(8,3) = T(2,9) $
92 T(6,1) = T(1,6) $ T(3,3) = T(1,6) $ T(6,2) = T(1,9) $
93 T(7,1) = T(1,7) $ T(2,4) = T(1,7) $ T(4,2) = T(1,7) $
94 T(8,1) = T(1,8) $ T(2,5) = T(1,8) $ T(3,4) = T(1,8) $ T(4,3) = T(1,8) $
95 T(9,1) = T(1,9) $ T(2,6) = T(1,9) $ T(3,5) = T(1,9) $ T(5,3) = T(1,9) $
96 T(10,1) = T(1,10) $ T(3,6) = T(1,10) $ T(6,3) = T(1,10) $
97 T(7,2) = T(2,7) $ T(4,4) = T(2,7) $ T(9,3) = T(2,10) $
98 T(8,2) = T(2,8) $ T(5,4) = T(2,8) $ T(3,7) = T(2,8) $ T(4,5) = T(2,8) $
99 T(9,2) = T(2,9) $ T(5,5) = T(2,9) $ T(3,8) = T(2,9) $ T(4,6) = T(2,9) $
100 T(10,2) = T(2,10) $ T(3,9) = T(2,10) $ T(5,6) = T(2,10) $ T(6,5) = T(2,10) $
101 T(5,7) = T(4,8) $ T(7,5) = T(4,8) $ T(8,4) = T(4,8) $
102 T(5,8) = T(4,9) $ T(7,6) = T(4,9) $ T(8,5) = T(4,9) $ T(9,4) = T(4,9) $
103 T(5,9) = T(4,10) $ T(8,6) = T(4,10) $ T(9,5) = T(4,10) $ T(10,4) = T(4,10) $
104 T(6,6) = T(3,10) $ T(10,3) = T(3,10) $
105 T(6,7) = T(4,9) $
106 T(6,8) = T(4,10) $
107 T(6,9) = T(5,10) $ T(9,6) = T(5,10) $ T(10,5) = T(5,10) $
108 T(8,7) = T(7,8) $
109 T(8,8) = T(7,9) $ T(9,7) = T(7,9) $
110 T(8,9) = T(7,10) $ T(9,8) = T(7,10) $ T(10,7) = T(7,10) $
111 T(9,9) = T(8,10) $ T(10,8) = T(8,10) $
112 T(10,6) = T(6,10) $
113 T(10,9) = T(9,10) $
114 COMMENT ASSIGN 3X3, 6X6, AND 10X10 PORTIONS OF T MATRIX TO NEW
115 MATRICES, TN(,), WHICH WILL BE USED IN SOLVING LINEAR EQUATIONS $
116 FOR I = (1,1,3) $ FOR J = (1,1,3) $ T3(I,J) = T(I,J) $
117 FOR I = (1,1,6) $ FOR J = (1,1,6) $ T6(I,J) = T(I,J) $
118 FOR I = (1,1,10) $ FOR J = (1,1,10) $ T10(I,J) = T(I,J) $
119 COMMENT ASSIGN 3, 6, AND 10 PORTIONS OF COLUMN VECTOR R TO NEW VECTORS $
120 FOR I = (1,1,3) $ R3(I) = R(I) $
121 FOR I = (1,1,6) $ R6(I) = R(I) $
122 FOR I = (1,1,10) $ R10(I) = R(I) $
123 COMMENT SOLVE LINEAR MATRIX EQUATIONS OF GENERAL FORM TS = R $
124 EXTERNAL PROCEDURE INPROD() $
125 PROCEDURE SOLV(N$A(,),B(,),Y())$SINGULAR)$
126 BEGIN
127 COMMENT THIS IS THE METHOD OF CROUT, WITH INTERCHANGES,
128 TO SOLVE AY=B FOR Y, GIVEN A AND B. $
129 COMMENT EXTERNAL PROCEDURE INPROD IS CALLED BY SOLV,
130 SO INPROD MUST BE AVAILABLE WHEN SOLV IS CALLED. $
131 COMMENT SINGULAR IS THE LABEL OF THE STATEMENT TO WHICH
132 SOLV() EXITS IF A() IS SINGULAR $
133 COMMENT REAL A(,),B(,),Y() $
134 INTEGER I,IMAX,J,K,N$
135 FOR K=(1,1,N)$
136 BEGIN
137 TEMP = 0$
138 FOR I=(K,1,N)$
139 BEGIN
140 XX = A(I,K) - INPROD(1,1,K-1,A(I,),A(,K))$
141 A(I,K) = XX $
142 IF ABS(XX) GTR TEMP $
143 BEGIN
144 TEMP = ABS(XX) $
145 IMAX = I
146 END

```

```

147     ENDS
148     IF TEMP EQL 0.0$
149     GO SINGULAR$
150     COMMENT WE HAVE FOUND THAT A(IMAX,K) IS THE LARGEST PIVOT IN COL K.
151     NOW WE INTERCHANGE ROWS K AND IMAX$
152     IF IMAX NEQ K$
153     BEGIN
154         FOR J = (1,1,N)$
155         BEGIN
156             TEMP=A(K,J)$ A(K,J)=A(IMAX,J)$ A(IMAX,J)=TEMP
157         ENDS
158         TEMP=B(K)$B(K)=B(IMAX)$B(IMAX)=TEMP
159     ENDS
160     COMMENT NOW FOR THE ELIMINATIONS
161     DENOM = A(K,K) $
162     FOR I=(K+1,1,N)$
163     A(I,K) = A(I,K)/DENOM $
164     FOR J = (K+1,1,N)$
165     A(K,J) = A(K,J) - INPROD(1,1,K-1,A(K, ),A(,J)) $
166     B(K) = B(K) - INPROD(1,1,K-1,A(K, ),B( ))
167     ENDS
168     FOR I=(1,1,N)$ Y(I) = A(I,I)$
169     COMMENT NOW FOR THE BACK SUBSTITUTION$
170     FOR K = (N,-1,1)$
171     Y(K) = (B(K) - INPROD(K+1,K+1,N-K,A(K, ),Y( )))/A(K,K) $
172     RETURN
173 END SOLV($
174 SOLV( 3$ T3(, ), R3(, ), S3( ) $ MAT6) $
175 MAT6..
176 SOLV( 6$ T6(, ), R6(, ), S6( ) $ MAT10)$
177 MAT10..
178 SOLV(10$T10(, ),R10(, ),S10( ) $ START)$
179 COMMENT SUBSTITUTE X AND Y MAP VALUES IN EQUATIONS OF FITTED SURFACES $
180 FOR I =(1,1,N) $
181     BEGIN
182     X(I,4) = S3(1) + (S3(2).X(I,1)) + (S3(3).X(I,2)) $
183     X(I,5) = X(I,3) - X(I,4) $
184     X(I,6) = S6(1) + (S6(2).X(I,1)) + (S6(3).X(I,2)) +
185     (S6(4).(X(I,1)*2.0)) + (S6(5).(X(I,1).X(I,2))) +
186     (S6(6).(X(I,2)*2.0)) $
187     X(I,7) = X(I,3) - X(I,6) $
188     X(I,8) = S10(1) + (S10(2).X(I,1)) + (S10(3).X(I,2)) +
189     (S10(4).(X(I,1)*2.0)) + (S10(5).(X(I,1).X(I,2))) +
190     (S10(6).(X(I,2)*2.0)) + (S10(7).(X(I,1)*3.0)) +
191     (S10(8).(X(I,1)*2.0).X(I,2)) + (S10(9).(X(I,1).X(I,2)*2.0))
192     +(S10(10).(X(I,2)*3.0)) $
193     X(I,9) = X(I,3) - X(I,8) $
194     END $
195 COMMENT COMPUTE ERROR MEASURE AND PERCENT TOTAL SUM OF SQUARES FOR EACH
196 TREND SURFACE $
197 EM1 = EM2 = EM3 = 0.0 $
198 FOR I =(1,1,N) $
199     BEGIN
200     EM1 = EM1 +(X(I,5)*2.0) $
201     EM2 = EM2 +(X(I,7)*2.0) $
202     EM3 = EM3 +(X(I,9)*2.0) $
203     END $
204     E1 = EM1/(N-1) $
205     E2 = EM2/(N-1) $
206     E3 = EM3/(N-1) $
207 SMLN = LNSQ = SMQD = QDSQ = SMCB = CBSQ = SMZ = ZSQ = 0.0 $
208 FOR I =(1,1,N) $
209     BEGIN
210     SMLN = SMLN + X(I,4) $
211     LNSQ = LNSQ + (X(I,4)*2.0) $
212     SMQD = SMQD + X(I,6) $
213     QDSQ = QDSQ + (X(I,6)*2.0) $
214     SMCB = SMCB + X(I,8) $
215     CBSQ = CBSQ + (X(I,8)*2.0) $
216     SMZ = SMZ + X(I,3) $
217     ZSQ = ZSQ + (X(I,3)*2.0) $
218     END $
219 ZOR = ZSQ - ((SMZ*2.0)/(N-1)) $
220 PTS1 = 100.0((LNSQ -((SMLN*2.0)/(N-1)))/ZOR)$
221 PTS2 = 100.0((QDSQ -((SMQD*2.0)/(N-1)))/ZOR)$
222 PTS3 = 100.0((CBSQ -((SMCB*2.0)/(N-1)))/ZOR)$

```

```

223 QDON = QDSQ - LNSQ $ CBON = CBSQ - QDSQ $
224 OUTPUT ERMA(E1,E2,E3,PTS1,PTS2,PTS3)$
225 FORMAT FMTR(*ERROR MEASURE LINEAR TREND SURFACE = *,X31.2,W,W,
226 *ERROR MEASURE QUADRATIC TREND SURFACE = *,X28.2,W,W,
227 *ERROR MEASURE CUBIC TREND SURFACE = *,X32.2,W,W,
228 *PERCENT TOTAL SUM SQUARES LINEAR SURFACE = *,X25.2,W,W,
229 *PERCENT TOTAL SUM SQUARES QUADRATIC SURFACE = *,X22.2,W,W,
230 *PERCENT TOTAL SUM SQUARES CUBIC SURFACE = *,X26.2,W,W)$
231 OUTPUT VAR(LNSQ, EM1, QDSQ, QDON, EM2, CBSQ, EM3, CBON ) $
232 FORMAT FVAR(*SUM OF SQUARES DUE LINEAR COMPONENT = *,X30.2,W4,W,
233 *SUM OF SQUARED DEVIATIONS FROM LINEAR = *,X28.2,W,W,
234 *SUM OF SQUARES DUE LINEAR + QUADRATIC COMPONENT = *,X18.2,
235 W,W,*SUM OF SQUARES DUE TO QUADRATIC ALONE = *,X28.2,W,W,
236 *SUM OF SQUARED DEVIATIONS FROM LINEAR + QUADRATIC = *,
237 X16.2,W,W,
238 *SUM OF SQUARES DUE LINEAR+QUADRATIC+CUBIC = *,X24.2,W,W,
239 *SUM OF SQUARED DEVIATIONS FROM LINEAR+QUADRATIC+CUBIC = *,
240 X12.2,W,W,
241 *SUM OF SQUARES DUE CUBIC ALONE = *,X35.2,W, W)$
242 AZM = R(1)/N $
243 OUTPUT ALPHA(A1,A2,A3,A4,A5,A6,A7,A8,A9,A10,A11,A12)$
244 OUTPUT ODS1(FOR I =(1,1,N) $ FOR J =(1,1,9) $ X(I,J)) $
245 FORMAT FTPC(*5$,12A6,$$,W)$ WRITE ($$ALPHA,FTPC) $
246 FORMAT FMTA(12A6,W3,W) $
247 OUTPUT C03(S3(1),S3(2),S3(3)),
248 C06(S6(1),S6(2),S6(3),S6(4),S6(5),S6(6)),
249 C10(S10(1),S10(2),S10(3),S10(4),S10(5),S10(6),S10(7),S10(8),
250 S10(9),S10(10)) $
251 FORMAT CF3(W,*EQUATION COEFFICIENTS *$,W,*LINEAR, Z = *,X12.5,***,
252 X12.5,**X +*,X12.5,*Y*,W,W),
253 CF6(*LIN + QUAD, Z = *,X12.5,***,X12.5,**X +*,X12.5,*Y +*,
254 X12.5,**X2 +*,X12.5,*XY +*,X12.5,*Y2*,W,W),
255 CF10(*LIN + QUAD + CUB, Z = *,X12.5,***,X12.5,**X +*,X12.5,*Y +*,
256 X12.5,**X2 +*,X12.5,*XY +*,X12.5,*Y2*,W,B10,***,X12.5,**X3 +*,
257 X12.5,**X2Y +*,X12.5,*XY2 +*,X12.5,*Y3*,W,W)$
258 FORMAT HED(* XCOORD YCOORD Z-VALUE 1ST-TRD 1ST-RSD 2ND-TRD 2ND-RSD*,
259 * 3RD-TRD 3RD-RSD*,W,W) $
260 FORMAT FMT1(*5$,9X8.1,W) $
261 WRITE ($$ ALPHA, FMTA) $
262 FORMAT JIM(*10 X 10 (X,Y) MATRIX VALUES *,W,W) $ WRITE ($$ JIM)$
263 OUTPUT TRAY(FOR I=(1,1,10)$ FOR J=(1,1,10) $ T(I,J))$
264 FORMAT FTRA(W,10F12.4,W) $
265 WRITE ($$ TRAY,FTRA) $
266 FORMAT JOE(W,*1 X 10 COLUMN VECTOR VALUES*,W,W)$ WRITE ($$ JOE)$
267 OUTPUT RAR(FOR I =(1,1,10)$ R(I))$ WRITE ($$ RAR, FTRA) $
268 WRITE ($$ C03, CF3) $ WRITE ($$ C06, CF6)$ WRITE ($$ C10, CF10) $
269 WRITE ($$ ERMA, FMTR ) $ WRITE ($$ VAR, FVAR) $
270 COMMENT CALCULATE VOLUME WITHIN MAP AREA BENEATH EACH OF THE THREE
271 TREND SURFACES BY INTEGRATING IN RESPECT TO BOTH X AND Y, AND EVALUA-
272 TING INTEGRAL EMPLOYING VALUES FOR MAP BOUNDARIES $
273 XD1 = XR - XL $ XD2 = (XR.XR) - (XL.XL) $
274 XD3 = (XR.(XR.XR)) - (XL.(XL.XL)) $
275 XD4 = ((XR.XR).(XR.XR)) - ((XL.XL).(XL.XL)) $
276 YD1 = YB - YT $ YD2 = (YB.YB) - (YT.YT) $
277 YD3 = (YB.(YB.YB)) - (YT.(YT.YT)) $
278 YD4 = ((YB.YB).(YB.YB)) - ((YT.YT).(YT.YT)) $
279 VLN = (((S3(1).XD1) + ((S3(2).XD2)/2)).YD1) + (((S3(3).XD1).YD2)/2) $
280 VQD = (((S6(1).XD1) + ((S6(2).XD2)/2) + ((S6(4).XD3)/3)).YD1)
281 + (((S6(3).XD1)/2) + ((S6(5).XD2)/4)).YD2)
282 + (((S6(6).XD1).YD3)/3) $
283 VCB = (((S10(1).XD1) + ((S10(2).XD2)/2) + ((S10(4).XD3)/3)
284 + ((S10(7).XD4)/4)).YD1) + (((S10(3).XD1)/2)
285 + ((S10(5).XD2)/4) + ((S10(8).XD3)/6)).YD2)
286 + (((S10(6).XD1)/3) + ((S10(9).XD2)/6)).YD3)
287 + (((S10(10).XD1)/4).YD4) $
288 AR = YD1.XD1 $
289 AZ1 = VLN/AR $ AZ2 = VQD/AR $
290 AZ3 = VCB/AR $
291 OUTPUT VOL(VLN, VQD, VCB,AZM, AZ1, AZ2, AZ3, AR) $
292 FORMAT FMVL(W,*VOLUME BENEATH LINEAR SURFACE =*,B8,X12.2,W,W,
293 *VOLUME BENEATH LIN+QUAD SURFACE =*,B6,X12.2,W,W,
294 *VOLUME BENEATH LIN+QUAD+CUB SURFACE =*,B2,X12.2,W,W,
295 *ARITH. MEAN Z, = SUM OF Z VALUES/ N = *,X12.2,W,W,
296 *AVERAGE Z VALUE, LINEAR SURFACE =*,B6,X12.2,W,W,
297 *AVERAGE Z VALUE, LIN+QUAD SURFACE =*,B4,X12.2,W,W,
298 *AVERAGE Z VALUE, LIN+QUAD+CUB SURFACE =*,X12.2,W,W,
299 *AREA OF MAP IN SQUARED UNITS*,B11,X12.2,W,W) $

```

```

300 WRITE ($$ VOL, FMVL) $
301 WRITE ($$ HED) $
302 WRITE ($$ ODS1, FMT1) $
303 COMMENT CALCULATE DX AND DY, AND SUBSTITUTE PROGRESSIVELY INCREASING
304 VALUES OF X AND Y MAP VALUES IN FITTING EQUATIONS AND CONTOUR $
305 DX = (XR - XL)/HOR $
306 DY = (YB - YT)/VERT $
307 OUTPUT CNDATA(XL, XR, YT, YB, RF, CON) $
308 FORMAT CONDAT(*X VALUE LEFT EDGE OF MAP = *,X9.1,
309 * X VALUE RIGHT EDGE OF MAP = *,X8.1,
310 * Y VALUE TOP EDGE OF MAP = *,X8.1,W,
311 * Y VALUE BOTTOM EDGE OF MAP = *,X7.1,
312 * REFERENCE CONTOUR VALUE = *,X9.1,
313 * CONTOUR INTERVAL = *,X16.1,W,W)$
314 W = 1 $
315 CALZ1..
316 I = 1 $
317 CALZ2..
318 IF W GEQ 4 $ GO RFADY $
319 IF W EQL 1 $
320 BEGIN
321 WRITE ($$ ALPHA, FMTA) $
322 FORMAT ZHED1(*CONTOURS OF LINEAR TREND SURFACE*,W) $
323 WRITE ($$ ZHED1) $
324 WRITE ($$ CNDATA,CONDAT) $
325 CALZ3..
326 BQX = S3(2).DX $
327 BQ = S3(1) + (S3(2).XL) + (S3(3).(YT + (DY.I))) $
328 FOR A = (1.0,1.0,HAR) $
329 Z(A) = BQ + (A.BQX) $
330 GO CONTOUR $
331 END $
332 IF W EQL 2 $
333 BEGIN
334 WRITE ($$ ALPHA, FMTA) $
335 FORMAT ZHED2(*CONTOURS OF LINEAR + QUADRATIC TREND SURFACE*,W)$
336 WRITE ($$ ZHED2) $
337 WRITE ($$ CNDATA,CONDAT) $
338 BS1 = S6(2).DX $
339 BS2 = S6(4).(2XL.DX) $
340 BS3 = S6(4).(DX.DX) $
341 CALZ4..
342 BSY = YT + (DY.I) $
343 BS4 = S6(5).(BSY.DX) $
344 BS = S6(1) + (S6(2).XL) + (S6(3).BSY) + (S6(4).(XL.XL)) +
345 (S6(5).(BSY.XL)) + (S6(6).(BSY.BSY)) $
346 BSA = BS1 + BS2 + BS4 $
347 FOR A = (1.0,1.0,HAR) $
348 Z(A) = (BSA.A) + (BS3.(A.A)) + BS $
349 GO CONTOUR $
350 END $
351 IF W EQL 3 $
352 BEGIN
353 WRITE ($$ ALPHA, FMTA) $
354 FORMAT ZHED3(*CONTOURS OF LINEAR + QUADRATIC + CUBIC TREND *,
355 *SURFACE*,W) $
356 WRITE ($$ ZHED3) $
357 WRITE ($$ CNDATA,CONDAT) $
358 BT1 = S10(2).DX $
359 BT2 = (2.DX).(XL.S10(4)) $
360 BT4 = S10(7) .(3DX(XL.XL)) $
361 BT5 = S10(7).(3XL.(DX.DX)) $
362 BT6 = S10(7).(DX.(DX.DX)) $
363 BT10=S10(4).(DX.DX) $
364 CALZ5..
365 BTY = YT + (DY.I) $
366 BT3 = S10(5).(BTY.DX) $
367 BT7 = S10(8).(2XL.(DX.BTY)) $
368 BT8 = S10(8).(BTY.(DX.DX)) $
369 BT9 = S10(9).(DX.(BTY.BTY)) $
370 BT = S10(1) + (S10(2).XL) + (S10(3).BTY) + (S10(4).(XL.XL)) +
371 (S10(5).(BTY.XL)) + (S10(6).(BTY.BTY)) + (S10(7).(XL.(XL.XL))) +
372 (S10(8).(BTY.(XL.XL))) + (S10(9).(XL.(BTY.BTY))) +
373 (S10(10).(BTY.(BTY.BTY))) $
374 BTA = BT1 + BT2 + BT3 + BT4 + BT7 + BT9 $
375 BTAA = BT5 + BT8 + BT10 $
376 FOR A = (1.0,1.0,HAR) $

```

```

377      Z(A) = (BTA.A) + (BTAA.(A.A)) + (BT6.(A.(A.A))) + BT $
378      GO CONTOUR $
379      END $
380 CONTOUR..
381 FOR J =(1,1,HOR) $
382      BEGIN
383      IF Z(J) LSS RF $
384          BEGIN
385          CV(J) = ARMIN(MOD(FIX((RF-Z(J))/CON),40) +1) $
386          GO THERE $
387          END $
388          CV(J) = ARPLS(MOD(FIX((Z(J)-RF)/CON),40)+1) $
389 THERE.. END $
390 OUTPUT ODCV(FOR J =(1,1,HOR) $ CV(J))$
391 FORMAT FTCV(132A1,W) $
392 WRITE ($$ ODCV,FTCV) $
393 I = I + 1 $
394 IF (W EQL 1) AND (I LEQ VERT) $ GO CALZ3 $
395 IF (W EQL 2) AND (I LEQ VERT) $ GO CALZ4 $
396 IF (W EQL 3) AND (I LEQ VERT) $ GO CALZ5 $
397 W = W + 1 $ GO CALZ1 $
398 READY..
399 OUTPUT OUT(FOR I =(1,1,K) $ PRINT(2,I)) $
400 FOR I =(1,1,N) $
401      BEGIN
402          IY(I) = X(I,2)/DY $
403          IX(I) = X(I,1)/DX $
404      END $
405 FOR Q =(3,2,9) $
406 BFGIN
407 WRITE ($$ ALPHA,FMTA)$ WRITE ($$ PLOT, PLFT) $
408 FOR LINE =(1,1,VERT) $
409      BEGIN
410          K=0$
411          FOR J =(1,1,N) $
412              BEGIN
413                  IF IY(J) EQL LINE $
414                      BEGIN
415                          K = K + 1 $
416                          PRINT(1,K) = IX(J) $
417                          PRINT(2,K) = X(J,Q) $
418                          DIGITS(K)= 0.4343.LOG((QQQ= ABS(X(J,Q))) +(QQQ LSS 1.0
419                              ))+ 2 $
420                      END $
421                  END $
422          FOR I =(2,1,K) $ FOR J =(1,1,I-1)$
423              BEGIN
424                  IF PRINT(1,I) LSS PRINT(1,J) $
425                      BEGIN
426                          FOR L = 1,2 $
427                              BEGIN
428                                  TEMP = PRINT(L,J) $
429                                  PRINT(L,J) = PRINT(L,I) $
430                                  PRINT(L,I) = TEMP $
431                              END $
432                                  TEMP = DIGITS(J)$ DIGITS(J)=DIGITS(I)$
433                                  DIGITS(I) = TEMP $
434                              END $
435                          END $
436          FOR I = (2,1,K) $ PRINT(3,I) = 0 $
437          PRINT(3,1) = PRINT(1,1) - DIGITS(1) $
438          FOR I = (2,1,K) $
439              BEGIN
440                  IF PRINT(3,I-1) LSS 0 $
441                      BEGIN
442                          PRINT(3,I) = PRINT(3,I-1) $
443                          PRINT(3,I-1) = 0 $
444                      ENDS
445                  PRINT(3,I) = PRINT(1,I) + PRINT(3,I) - .PRINT(1,I-1)
446                  - DIGITS(I) $
447              ENDS
448          WRITE ($$ OUT, FORM) $
449      END $
450 END $
451 GO START $ FINISH $
452 1 ((G 7 ( )G
453 J ,X)*PE.2 9 '76'75'7447 '7X'76 7 77X'75 449' 7448' '4 ' ' 9' '4 ' ' 9' 7
454 J7((#P'- 74'G 04'9 '76 75 6 '06705'-4'76 75 74 549
455 FINISH $

```

- (23) Evaluate data values with respect to reference contour and contour interval, and express value as an appropriate character as assigned in alphanumeric arrays ARMIN and ARPLS: lines 380-397.
- (24) Plot original data and residual values of first-, second-, and third-degree surfaces on maps that are same scale as contoured trend-surface maps. Data are sorted first according to Y coordinate values, and second according to X coordinate values: lines 398-450.
- (25) Go to START to read in new set of data, if present, or terminate use of program if no more data sets are available: line 451.
- (26) Binary external procedure INPROD used in matrix inversion procedure SOLV: lines 452-455.

Please note that on the actual program cards, a 2 should be punched in column 1, and the line numbers should be placed within columns 73-80.

#### INPUT TO TREND-SURFACE PROGRAM

Alphanumeric heading card.--The first card of each set of data must be a card containing alphabetical and numerical (alphanumeric) information for identification purposes. This information will be reproduced at the top of maps (Fig. 2) and other pages (Table 2) printed by the computer's printer. The card must be punched as follows:

- (1) 5 in column 1.
- (2) \$ in columns 2 and 75.
- (3) Any desired combination of letters, numbers, characters, and blanks in columns 3 through 74.

Program-control card.--The second card that accompanies each data set must be a program-control card containing the following information, in the order listed below. Note that each number is to be separated from adjacent numbers by at least one blank space on the card.

- (1) 5 in column 1.
- (2) An integer specifying whether the data are in integer or decimal form. Punch 2222 if data are integer; punch 4444 if data are decimal. (OP in program).
- (3) An integer specifying the number of data points (N).
- (4) An integer specifying the number of printed rows that map is to contain (VERT).
- (5) An integer specifying the number of columns in maps that are to be printed (HOR). Note that there are 10 columns and 6 rows to the inch. Appropriate numbers for VERT and HOR must be chosen so that maps printed by computer are not distorted in respect to the ratio of length to width. The maximum number of columns is 132 (assuming that an IBM 1403 printer is used).
- (6) A decimal number specifying the coordinate value of the right edge of the map (XR). Note that all coordinate values specifying edges of maps to be printed must be in the same units as X and Y coordinate values of data points.
- (7) A decimal number specifying coordinate value of left edge of map (XL).

- (8) A decimal number specifying coordinate value of bottom edge of map (YB).
- (9) A decimal number specifying coordinate value of top edge of map (YT).
- (10) A decimal number specifying the reference contour value (RF).
- (11) A decimal number specifying the contour interval (CON).

Data cards. --Data cards follow the control card. Each data card should have 5 punched in column 1. Data values must be separated by at least one space, and any convenient number of data values may be placed on a card, columns 2 through 80 being available. In BALGOL, format input specifications for data are not used. Each data point has three values which must be listed in the following order:

- (1) X or horizontal coordinate value.
- (2) Y or vertical coordinate value.
- (3) Z value.

In reading the data from the cards, the computer will read the numbers in groups of three. It is essential that the number of data points (three numbers per point) be exactly equal to the number N specified on the control card. The maximum number of data points (Z values) specified is 500, although this limit can be readily increased with very minor changes in the program.

The X and Y coordinate values may be in any convenient units. I have found tenths of an inch to be convenient, but miles, feet, kilometers, or any unit may be used. In order to keep all values positive, it is suggested that the coordinate origin be placed in the upper left corner of the map. The location of the origin is arbitrary and may be either at the corner of the area to be mapped or outside it. The X or horizontal coordinate values have positive signs and increase moving toward the right from the coordinate origin. The Y or vertical coordinate values have positive signs and increase moving downward from the coordinate origin. The control points may be irregularly spaced.

#### OUTPUT FROM TREND-SURFACE PROGRAM

X-Y matrix values. --Examples of output from the trend-surface program are shown in Table 2 and Figure 2. At the top of Table 2 are shown the elements of the 10 x 10 matrix used in solving the normal equations. Although the values of the matrix elements are not usually of interest in themselves, it may be helpful to scan them to insure that the capacity of the computer has not been exceeded if very large numbers are being used in the original data.

Equation constants. --The constants of the equations that describe first-, second-, and third-degree surfaces are printed out. The algebraic signs of constants are included.

Statistical measures of surfaces. --The following statistical measures of the trend surfaces and their residuals are calculated:

- (1) Error measure.
- (2) Percent of total sum of squares of each surface.
- (3) Sum of squares due to linear, quadratic, and cubic components, separately and combined, and sum of squared deviations from the first-, second-, and third-degree surfaces.

Table 2. --Example of output of tabular data from trend-surface program.

```

BOUGUER GRAVITY ANOMALIES IN NORTHWESTERN KANSAS - CI = 10 MG
10 X 10 (X,Y) MATRIX VALUES

  1.000, 02  3.841, 02  4.398, 02  2.038, 03  1.687, 03  2.591, 03  1.224, 04  8.916, 03  9.952, 03  1.709, 04
  3.841, 02  2.038, 03  1.687, 03  1.224, 04  8.916, 03  9.952, 03  7.848, 04  5.309, 04  5.257, 04  6.572, 04
  4.398, 02  1.687, 03  2.591, 03  8.916, 03  9.952, 03  1.709, 04  5.309, 04  5.257, 04  6.572, 04  1.196, 05
  2.038, 03  1.224, 04  8.916, 03  7.848, 04  5.309, 04  5.257, 04  5.244, 05  3.371, 05  3.120, 05  3.473, 05
  1.687, 03  8.916, 03  9.952, 03  5.309, 04  5.257, 04  6.572, 04  3.371, 05  3.120, 05  3.473, 05  4.600, 05
  2.591, 03  9.952, 03  1.709, 04  5.257, 04  6.572, 04  1.196, 05  3.120, 05  3.473, 05  4.600, 05  8.702, 05
  1.224, 04  7.848, 04  5.309, 04  5.244, 05  3.371, 05  3.120, 05  3.603, 06  2.228, 06  1.971, 06  2.056, 06
  8.916, 03  5.309, 04  5.257, 04  3.371, 05  3.120, 05  3.473, 05  2.228, 06  1.971, 06  2.056, 06  2.431, 06
  9.952, 03  5.257, 04  6.572, 04  3.120, 05  3.473, 05  4.600, 05  1.971, 06  2.056, 06  2.431, 06  3.338, 06
  1.709, 04  6.572, 04  1.196, 05  3.473, 05  4.600, 05  8.702, 05  2.056, 06  2.431, 06  3.338, 06  6.499, 06

1 X 10 COLUMN VECTOR VALUES

-1.029, 04 -3.529, 04 -4.479, 04 -1.772, 05 -1.551, 05 -2.610, 05 -1.029, 06 -7.828, 05 -9.061, 05 -1.707, 06

EQUATION COEFFICIENTS
LINEAR, Z = -134.90545+ 7.51003X + .72270Y
LIN + QUAD, Z = -139.28436+ 11.75949X + -1.77729Y + -.29449X2 + -.45042XY + .48825Y2
LIN + QUAD + CUB, Z = -137.15641+ 12.71459X + -7.19410Y + -.52583X2 + -.93808XY + 2.49575Y2
+ .04022X3 + -.05862X2Y + .10580XY2 + -.18304Y3

ERROR MEASURE LINEAR TREND SURFACE = 50.85
ERROR MEASURE QUADRATIC TREND SURFACE = 34.15
ERROR MEASURE CUBIC TREND SURFACE = 24.83
PERCENT TOTAL SUM SQUARES LINEAR SURFACE = 80.93
PERCENT TOTAL SUM SQUARES QUADRATIC SURFACE = 87.19
PERCENT TOTAL SUM SQUARES CUBIC SURFACE = 90.69

SUM OF SQUARES DUE LINEAR COMPONENT = 1090505.40
SUM OF SQUARED DEVIATIONS FROM LINEAR = 5034.33
SUM OF SQUARES DUE LINEAR + QUADRATIC COMPONENT = 1092158.80
SUM OF SQUARES DUE TO QUADRATIC ALONE = 1653.39
SUM OF SQUARED DEVIATIONS FROM LINEAR + QUADRATIC = 3380.91
SUM OF SQUARES DUE LINEAR+QUADRATIC+CUBIC = 1093081.61
SUM OF SQUARED DEVIATIONS FROM LINEAR+QUADRATIC+CUBIC = 2457.78
SUM OF SQUARES DUE CUBIC ALONE = 922.81

VOLUME BENEATH LINEAR SURFACE = -6798.72
VOLUME BENEATH LIN+QUAD SURFACE = -6806.03
VOLUME BENEATH LIN+QUAD+CUB SURFACE = -6809.02
ARITH. MEAN Z, = SUM OF Z VALUES/ N = -102.88
AVERAGE Z VALUE, LINEAR SURFACE = -102.54
AVERAGE Z VALUE, LIN+QUAD SURFACE = -102.65
AVERAGE Z VALUE, LIN+QUAD+CUB SURFACE = -102.70
AREA OF MAP IN SQUARED UNITS 66.30


XCOORD YCOORD Z-VALUE 1ST-TRD 1ST-RSD 2ND-TRD 2ND-RSD 3RD-TRD 3RD-RSD
5 5.2 1.7 -93.0 -94.6 1.6 -91.7 -1.3 -94.9 1.9
5 7.8 7.1 -77.0 -71.2 -5.8 -78.4 1.4 -77.3 .3
5 4.7 5.0 -102.0 -96.0 -6.0 -97.8 -4.2 -97.4 -4.6
5 2.1 5.2 -109.0 -115.4 6.4 -116.8 7.8 -113.6 4.6
5 7.3 1.2 -71.0 -79.2 8.2 -74.5 3.5 -72.9 1.9
5 4.8 5.9 -100.0 -94.6 -5.4 -95.9 -4.1 -93.8 -6.2
5 .6 3.6 -138.0 -127.8 -10.2 -133.4 -4.6 -133.1 -4.9
5 3.0 3.5 -109.0 -109.8 .8 -111.6 2.6 -112.9 3.9
5 1.7 3.8 -124.0 -119.4 -4.6 -122.8 -1.2 -122.3 -1.7
5 3.6 4.9 -95.0 -104.3 9.3 -105.7 10.7 -104.3 9.3
5 6.1 2.5 -92.0 -87.3 -4.7 -86.8 -5.2 -91.0 -1.1
5 2.3 6.3 -98.0 -113.1 15.1 -112.1 14.1 -108.1 10.1
5 3.3 4.1 -104.0 -107.2 3.2 -108.9 4.9 -109.1 5.1
5 7.8 1.1 -70.0 -75.5 5.5 -70.7 .7 -67.0 -3.0
5 2.0 1.1 -120.0 -119.1 -.9 -119.3 -.7 -120.7 -.7
5 5.0 8.1 -95.0 -91.5 -3.5 -88.5 -6.5 -88.7 -6.3
5 3.8 5.6 -100.0 -102.3 2.3 -103.1 3.1 -100.5 .5
5 4.5 2.8 -110.0 -99.1 -10.9 -99.2 -10.8 -102.9 -7.1
5 1.9 5.5 -116.0 -116.7 .7 -117.7 1.7 -114.0 -2.0
5 2.4 .1 -115.0 -116.8 1.8 -113.0 -2.0 -110.1 -4.9
5 6.1 2.0 -88.0 -87.6 -.4 -85.6 -2.4 -89.1 1.1
5 4.3 4.1 -99.0 -99.6 .6 -101.2 2.2 -102.5 3.5
5 4.8 1.0 -101.0 -98.1 -2.9 -93.1 -7.9 -94.0 -7.0
5 4.1 7.6 -92.0 -98.6 6.6 -95.4 3.4 -93.6 1.6
5 1.9 4.8 -124.0 -117.2 -6.8 -119.4 -4.6 -116.8 -7.2
5 4.8 7.4 -85.0 -93.5 8.5 -92.0 7.0 -90.0 5.0
5 6.4 7.7 -81.1 -81.3 .2 -83.0 1.9 -82.3 1.2
5 6.6 8.4 -77.0 -79.3 2.3 -79.9 2.9 -81.6 4.6
5 2.8 1.2 -112.0 -113.0 1.0 -111.6 -.4 -113.4 1.4
5 2.8 6.2 -100.0 -109.4 9.4 -108.7 8.7 -104.8 4.8
5 7.0 7.4 -81.0 -77.0 -4.0 -81.1 .1 -80.2 -.8
5 7.6 3.3 -79.0 -75.4 -3.6 -78.8 -.2 -82.3 3.3
5 6.9 8.4 -75.0 -77.0 2.0 -78.7 3.7 -80.4 5.4
5 5.9 6.5 -90.0 -85.9 -4.1 -88.4 -1.6 -86.6 -3.4

```

STRUCTURE TOP STONE CORRAL (PERMIAN) ROCKS IN WESTERN KANSAS - CI=100 FT

CONTOURS OF LINEAR + QUADRATIC TREND SURFACE


X VALUE LEFT EDGE OF MAP = .0 X VALUE RIGHT EDGE OF MAP = 171.0 Y VALUE TOP EDGE OF MAP = .0  
 Y VALUE BOTTOM EDGE OF MAP = 184.0 REFERENCE CONTOUR VALUE = 500.0 CONTOUR INTERVAL = 100.0



A

STRUCTURE TOP STONE CORRAL (PERMIAN) ROCKS IN WESTERN KANSAS - CI=100 FT

2ND ORDER RESIDUAL



B

Figure 2. Examples of printed maps produced as output from trend-surface program: A, contours on second-degree trend surface; B, residual values from second-degree trend surface shown in A.

- (4) Volume beneath first-, second-, and third-degree surfaces within limits of mapped area.
- (5) Average Z values based on arithmetic mean, and spatially weighted values beneath first-, second-, and third-degree surfaces.
- (6) Mapped area in square units.

For a thorough discussion of the mathematical principles involved in computation and application of these statistical methods see Harbaugh (in press). The method of calculating volumes and spatially weighted averages has been adapted from Whitten (1962).

Values at individual data points.--The following values of individual data points are printed out and may also be simultaneously punched on cards if desired by changing the W in line 260 to C.

- (1) X or horizontal coordinate value.
- (2) Y or vertical coordinate value.
- (3) Z value.
- (4) Value of first-degree trend surface at that point.
- (5) Value of first-degree residual.
- (6) Second-degree trend value.
- (7) Second-degree residual value.
- (8) Third-degree trend value.
- (9) Third-degree residual value.

Maps.--The following maps are printed out:

- (1) First-degree trend map.
- (2) Second-degree trend map (Fig. 2).
- (3) Third-degree trend map.
- (4) Map on which Z values are plotted.
- (5) Map of first-degree residuals.
- (6) Map of second-degree residuals (Fig. 2).
- (7) Map of third-degree residuals.

Identifying information is printed out at the top of each map. Alternating bands of printed characters and blanks represent contours on the trend maps and permit each contour line to be identified. The printed characters used are listed in Table 3. The printed characters have been arranged so that there is little likelihood of ambiguity. The steps in calculating the values of contour lines are as follows:

- (1) Determine the number of contour intervals above or below the reference contour as represented by the bands of characters or blanks referred to in Table 3.
- (2) Multiply this number by the contour interval (declared on program control card, CON).
- (3) Add algebraically to the reference contour value. For example, if the reference contour value

is 100, and the contour interval is 10, then the contour value represented by the lower edge of the band printed with B's is 60. The reference contour value is marked by the edge of the band of \$'s which faces the band of A's.

Table 3. --List of characters that correspond with contour intervals of printed contour maps. Empty places in column indicate that no character has been printed.

Number of contour intervals above (+) or below (-) reference contour.	Character printed (or blank) in band, of which lower edge denotes contour value.	Number of contour intervals above (+) or below (-) reference contour.	Character printed (or blank) in band, of which lower edge denotes contour value.
-40	T	+ 1	
-39		+ 2	
-38	S	+ 3	1
-37		+ 4	
-36	R	+ 5	2
-35		+ 6	
-34	Q	+ 7	3
-33		+ 8	
-32	P	+ 9	4
-31		+10	
-30	O	+11	5
-29		+12	
-28	N	+13	6
-27		+14	
-26	M	+15	7
-25		+16	
-24	L	+17	8
-23		+18	
-22	K	+19	9
-21		+20	
-20	J	+21	0
-19		+22	
-18	I	+23	\$
-17		+24	
-16	H	+25	*
-15		+26	
-14	G	+27	-
-13		+28	
-12	F	+29	.
-11		+30	
-10	E	+31	+
- 9		+32	
- 8	D	+33	=
- 7		+34	
- 6	C	+35	W
- 5		+36	
- 4	B	+37	X
- 3		+38	
- 2	A	+39	Y
- 1			
0	\$		

#### REFERENCES

- Forgotson, J. M., Jr., 1963, How computers help find oil: *The Oil and Gas Journal*, v. 61, no. 11, p. 100-109.
- Harbaugh, J. W., in press, A computer method for four-variable trend analysis applied to crude oil gravity variations in southeastern Kansas: *Kansas Geol. Survey Bull.*
- Krumbein, W. C., 1959, Trend surface analysis of contour-type maps with irregular control-point spacing: *Jour. Geophys. Research*, v. 64, no. 7, p. 823-834.
- Merriam, D. F., and Harbaugh, J. W., in press, Computer analysis of regional and residual components of geologic structure in Kansas: *Kansas Geol. Survey Bull.*
- Whitten, E. H. T., 1962, A new method for determination of the average composition of a granite massif: *Geochimica et Cosmochimica Acta*, v. 26, p. 545-560.

KANSAS GEOLOGICAL SURVEY COMPUTER PROGRAM  
THE UNIVERSITY OF KANSAS, LAWRENCE

PROGRAM ABSTRACT

Title (If subroutine state in title):

BALGOL program for trend-surface mapping using an IBM 7090 computer

Computer: IBM 7090

Date: August 1963

Programming language: BALGOL

Author, organization: John W. Harbaugh, Department of Geology, Stanford University

Stanford, California

Direct inquiries to: author or to Daniel F. Merriam, State Geological Survey, Lawrence, Kansas

Name: John W. Harbaugh (author)

Address: Department of Geology, Stanford University

Stanford, California

Purpose/description: Fitting of first-, second-, and third-degree trend surfaces to geological and other data.

Accepts irregularly spaced data points. Automatically contours trend surfaces and plots original data as well as first-, second-, and third-degree residual values in a series of maps. In addition, calculates various statistical properties of trend and residual surfaces including (a) error measure, (b) sums of squares, (c) percent of total sums of squares, (d) volume beneath each trend surface, (e) spatially weighted averages, and (f) arithmetic mean.

Mathematical method: Generates matrix and vector; obtains constants of equations of trend surfaces by matrix inversion.

Restrictions, range: Present limit is 500 data points, but this can be increased substantially by changing the array dimensions.

Storage requirements:

Equipment specifications: (not pertinent, inasmuch as 7090's have 32,000+ words of high-speed memory)


Memory 20K \_\_\_\_\_ 40K \_\_\_\_\_ 60K \_\_\_\_\_ K \_\_\_\_\_

Automatic divide: Yes \_\_\_\_\_ No \_\_\_\_\_ Indirect addressing: Yes \_\_\_\_\_ No \_\_\_\_\_

Other special features required \_\_\_\_\_

Additional remarks (include at author's discretion: fixed/float, relocatability; optional: running time, approximate number of times run successfully, programming hours) Compiles from BALGOL source deck in 18 seconds; typical running time, assuming 200 data points and maps that contain 60 square inches, is about 30 seconds. The Stanford University Computation Center will make its BALGOL compiler tape available to any 7090 user, provided the user will send a tape from a 7090 tape unit to the Stanford Computation Center.







Special  
Distribution  
Publication

1. The Kansas mineral industry...1962; with directory of Kansas mineral producers, by Grace Muilenburg, R. G. Hardy, and Allison Hornbaker, 1963.
2. Economic development for Kansas, mineral and water resources: Report of the Governor's Economic Development Committee, by W. W. Hambleton, and others, 1962.
3. BALGOL program for trend-surface mapping using an IBM 7090 computer, by J. W. Harbaugh, 1963.